Available online at www.sciencedirect.com

ScienceDirect

Computers in Biology and Medicine 00 (2014) 1–11

CBM

Object-oriented modeling and simulation of hemodynamic in Modelica

Tomáš Kulhánek*, Jiří Kofránek, Marek Mateják

Institute of Pathological Physiology, Charles University in Prague, U Nemocnice 5, 12853 Praha 2, Czech Republic

Abstract

Selected models of the cardiovascular system are presented in Modelica language and an acausal concept of designing model is discussed. The described Modelica models and libraries are used in teaching of physiology for graduate medical students and in teaching of modeling and simulation for graduate biomedical engineering students.

Keywords: Object-oriented modeling, MODELICA programming language, OPENMODELICA modeling environment, DYMOLA modeling environment, Cardiovascular system

1. Introduction

The large scale mathematical description of cardiovascular circulation and it's regulation was introduced by Guyton et al. in 1972[1]. This development continues today by Hester et al. who introduced a complex model not limited to cardiovascular system - HumMod - a derivative of the Guyton's model and in-house modeling language and simulation tool[2, 3]. Burkhoff et al. use electrical analogy to describe hemodynamics of cardiovascular system [4, 5] and published recently an interactive educational book with simulator based on their model[6]. Van Meurs et al. use similar approach to describe utilizing hydraulic components[7, 8, 9] and their model of cardiorespiratory dynamic is implemented in the Human Patient Simulator (HPS®) and derived products.

It is not simple workflow to develop such model from conceptual design phase to implementation in a programming language and execution code in a simulator. Thus there is an effort to develop some standards and tools appropriate for physiological research. The NSR Physiome project introduced a JSIM Java based simulation system to support modeling in physiology and introduces a repository of several hundred of models[10]. The similar effort was done by IUPS Physiome project and it's XML based standard CellML, where tools and repository of models are presented[11].

There is also an effort to utilize some standards and tools used by science and industry in different domains. Fernandez de Canete et al. described a closed loop cardiovascular model and mechanism of arterial pressure control in MATLAB® based modeling module SIMSCAPE™[12] and recently in Modelica language and DYMOLA tool with an approach to express the process of computation the model state variables[13].

Kofranek et al. implemented Guyton's 1972 model in MATLAB® Simulink[14]. However, the complexity of the model increased from Guyton's model to HumMod and it becomes too complicated keeping the model up-to-date using causal - block oriented tools like MATLAB®Simulink. Therefore an implementation of the HumMod model was introduced in object oriented Modelica language[15][16]. Modelica language is standard modeling language developed by industry vendors together with academia mantained by the international Modelica association[17]. It introduces a acausal modeling technique, where causality is solved by the tool during compilation[18].

It is even much more difficult to reuse, enhance and integrate the models together or implement new knowledge to build comprehensive integrative model of human physiology. It seems that the approach of Hester et al. succeeded in this task [2, 3]. Based on the selected papers cited above and with experiences of developing the mentioned models we identified these critical problems when designing models of physiological systems for research purpose:

1. Type checking and units conversion - For practical reasons, physiological models use well defined, but dif-

*Corresponding author
Email address: tomas.kulhanek@lf1.cuni.cz (Tomáš Kulhánek)

ferent units for the same quantities. E.g. for pressure it is used mmHg, torr instead of SI unit Pascal. Mismatch in units used in equations and input parameters as well as mismatch in different types connected by a mistake cause errors during development or validation phase.

2. Understandability - A model implementation may contain many redundant components or equation which describe same or similar things. Some models converge to a set of many equations or components within single diagrams or module which describes several distinct types of behavior together.
3. Reproducibility - this is a key element in scientific research as any scientific result should be reproducible following the described methods.

The first problem can be addressed by the tools and libraries used for modeling. The other problem shows up on non-hierarchical or block oriented modeling languages. But it appears in object-oriented tools when using non-appropriate techniques as identified by Tiller as "kitchen-sink" and "DRY" antipatterns in Modelica[19]. The recommended best practice is to "divide and conquer" - build component models that attempt as much as possible to describe individual effects[19].

The third problem is sometimes hard to achieve as the models presented in scientific papers contains errors or are incomplete. It is a success of current physiome project namely jsim or cellml where repositories of the models contains a so-called curated models, which verifies the mathematical correctness and reproducibility of the published models in physiology by giving the models in a source codes together with tools to build and reproduce the results.

In this paper we introduce models of hemodynamics of cardiovascular system in the MODELICA modeling language with library for modeling physiology - PHYSIOLIBRARY. We follow object-oriented approach to prevent the problems mentioned above and introduce understandable, reusable and reproducible implementation of model of Meurs[9] and model of Burkhoff[5, 4]. We believe that this approach will gain significant impact for the current physiome projects with another approach using MODELICA modeling language to curate complex models in an integrative approach, which not all efforts mentioned above succeeded.

2. Methods

Each model in Modelica is presented by it's icon where input, output or acausal connectors are presented, and by the diagram or textually written equations. Each model is in object oriented terms a class, which can be used for object instatiation or inherited when defining a variant of the model. In text listing within this article we use "pseudocode" where we ommit library prefixes and annotations. We present the diagram view of the models as figures. Based on equations or diagrams and connections among different models, the tool generates the equations and tries to solve causality of them.

The open source OPENMODELICA tool developed by Open Source Modelica Consortium[20] can be freely utilized to model and simulate models presented within this paper.

Modelica library PHYSIOLIBRARY introduced by Matejak et al.[21] brings reusable generic components in different domains useful to develop multidomain models in physiology. Among all other things, it utilizes a feature of Modelica 3.3 specification and define new display units which are usually used to describe physiological quantities in scientific papers. The simulation is usually done in SI units, unit conversion from inputs and outputs is secured automatically by advanced modeling tool and saved into the model definition.

To model hemodynamic of cardiovascular system we can use these components from the hydraulic domain defined in Physiolibrary.Hydraulic.Components. For each component the following table contains icon - which can be utilized in further model diagram - and set of equations, these are defined implicitly, explicit solution is done by tool.

	Hydraulic Resistor is characterized in Physiolibrary by conductance which is reciprocal value of resistance (conductance = 1/resistance), electrical analogy is resistor. $q_{in}.q = c * (q_{in}.pressure - q_{out}.pressure)$
	Elastic compartment is characterized by elasticity, electrical analogy is capacity. Special case is elastic compartment with variable elasticity. $q_{in}.pressure = \frac{volume}{c}$ $\frac{dvolume}{dt} = q_{in}.q$
	Inertia element is characterized by inertance, and tries to describe the tendency of liquid to keep moving. Electric analogy is inductor.
	Valve is characterized by the direction where the flow is allowed.
	Hydraulic connectors are characterized by the flow rate and pressure, hydraulic analogy of "Kirchhof law" are kept.
	Hydraulic pump generates the desired flow in the direction
	Pressure sensor will detect the pressure within the flow and makes it as an output result
	Spline aproximation. It is characterized by empirically retrieved function defined by the data points. The missing points are computed during simulation.

Figure 1 is an example of PHYSIOLIBRARY using hydraulic components to express the Guyton-Coleman-



Figure 1. The part of cardiovascular system of Guyton-Coleman-Granger model from 1972[1] in OPENMODELICA. Both right heart and left heart are controlled by the sensing of the incoming pressure which via Frank Starling law influence the heart pump. Pulmonary and systemic circulation are characterized as serial set of elastic compartments and hydraulic resistors.



Figure 2. The conceptual model by Meurs[9] using Physiolibary components in OPENMODELICA.

Granger model of cardiovascular part separated from its regulation [1].

2.1. Hemodynamics by Meurs et al.

The conceptual model of Meurs[9] describes the parts of cardiovascular circulation as set of elastic compartments, valves, hydraulic resistor and inertia element. Pulsatile cardiovascular system needs to define also components for generating heart pulse and variable features of the system. If we implement the conceptual model in MODELICA using the components from Physiolibrary, the model could look as on figure 2, however, the subsystem of variable atrial/ventricular elasticity and heart intervals is not there. Adding them would complicate the diagram and lead to kitchen-sink antipattern which would cause problems on further reusability.

Thus we separate the model into three distinct and model each part separately as on figure 3.

The model of heart contains a component that generates heart intervals based on the current simulation time reffered as "time" in source code and heart period computed as reciprocal value from heart rate:



HeartIntervals component defines empirical set of equations in Modelica code, outputs the relative time of atrial and ventricular systole.

```
model HeartIntervals
  TimeOutput Tas "duration of atrial systole";
  TimeOutput Tav "atrioventricular delay";
  TimeOutput Tvs "duration of ventricular systole";
```



Figure 3. The top-level structure of the cardiovascular model.

```
  TimeOutput T0 "start time of cardiac cycle";
  FrequencyInput HR "in SI units (1/s = Hz)";
  discrete Modelica.SIunits.Time HP(start = 0)
    "heart period - duration of cardiac cycle";
  Boolean b(start = false);
equation
  b = time - pre(T0) >= pre(HP); //new pulse occurs
  when {initial(), b} then
    T0 = time;      //start time of cardiac cycle
    HP = 1 / HR;    //heart period in sec
    Tas = 0.03 + 0.09 * HP; //duration of a.systole
    Tav = 0.01;     //atrioventricular delay
    Tvs = 0.16 + 0.2 * HP; //duration of v.systole
  end when;
end HeartIntervals;
```

The variable atrial and ventricular elasticity is based on the heart intervals obtained from the previously defined component.

AtrialElastance component generates constant elastance or sinusoidal signal based on input times Tas and current time of simulation.

```
model AtrialElastance
  TimeInput Tas;
  TimeInput T0;
  HydraulicElastanceOutput Et "elasticity";
  parameter HydraulicElastance EMIN "Diastolic el.";
  parameter HydraulicElastance EMAX "Systolic el.";
equation
  if time - T0 < Tas then
    Et = EMIN + (EMAX - EMIN) * sin(
          Modelica.Constants.pi * (time - T0) / Tas
         );
  else
    Et = EMIN;
  end if;
end AtrialElastance;
```

VentricularElastance component generates constant elastance or sinusoidal signal based on input times Tas,Tav,Tvs and current time of simulation.

```
model VentricularElastance
...//TimeInput Tas,Tav,Tvs
  //parameters EMIN, EMAX
  TimeOutput HeartInterval;
  HydraulicElastanceOutput Et;
...
  constant Real Kn = 0.57923032735652;
  //... where the t * sin(pi*t) has its maximum
equation
  HeartInterval = time - T0;
  Et = EMIN + (EMAX - EMIN) * Et0;
  if HeartInterval >= Tas+Tav
  and HeartInterval < Tas+Tav+Tvs then
    Et0 = (HeartInterval - (Tas + Tav)) / Tvs * sin(
            Modelica.Constants.pi*(
              HeartInterval-(Tas+Tav)
            ) / Tvs
          ) / Kn;
  else
    Et0 = 0;
  end if;
end VentricularElastance;
```

A submodel of side of the heart (left or right) is in the diagram on the figure 4. It contains hydraulic connectors "inflow" and "outflow" which connects the blood flow via atrium through atrioventricular valve(tricuspid in right heart, bicuspid in left heart) to ventricle and via ventricular artery valve(pulmonic in right heart and aortic in left heart). The atrium and ventricle elasticities are influenced by variable elastance components defined before and by parameters defining unstressed volume VxAU, VxVU. The valves are characterized by conductance parameters CxABackflow CxVBackflow which are normally 0 and resistance parameters RxAOutflow RxVOutflow (replace x by L for left heart and by R for right heart).



Figure 4. Model of SideOfHeart component in OPENMODELICA.

The parameters for the left side of the heart and right side of the heart differs. We use inheritance and redeclare the parameter values.

LeftHeart inherits behavior, defines own values of parameters in Modelica text view in SI units.

```
model LeftHeart
  extends SideOfHeart(
    VxAU(k = 3e-005),
    VxVU(k = 6e-005),
    RxAOutflow(k = 399967.162245),
    RxVOutflow(k = 1066579.09932),
    ventricularElastance(
      EMIN = 11999014.86735,
      EMAX = 533289549.66),
    atrialElastance(
      EMIN = 15998686.4898,
      EMAX = 37330268.4762),
    atrium(volume_start = 4e-005),
    ventricle(volume_start = 0.00013));
  extends Physiolibrary.Icons.LeftHeart;
end LeftHeart;
```

RightHeart inherits behavior, defines own values of parameters in Modelica text view in SI units.

```
model RightHeart
  extends SideOfHeart(
   VxAU(k = 3e-005),
   VxVU(k = 4e-005),
   RxAOutflow(k = 399967.162245),
   RxVOutflow(k = 399967.162245),
```



Figure 5. Model of whole heart in OPENMODELICA.



Figure 6. Model of pulmonary circulation in OPENMODELICA.

```
    ventricularElastance(
      EMIN = 7599376.082655,
      EMAX = 65327969.83335),
    atrialElastance(
      EMIN = 6666119.37075,
      EMAX = 19998358.11225),
    atrium(volume_start = 4e-005),
    ventricle(volume_start = 0.00013));
  extends Physiolibrary.Icons.RightHeart;
end RightHeart;
```

The heart model (figure 5) connects left and right heart instances with a component of heart intervals generating the periodic time signals.

The model of pulmonary circulation 6 is set of elastic compartments and resistances. Systemic circulation 10 divides into three parts: systemic arteries, peripherals and veins on figures 7, 8, 9.

The complete model by Meurs on figure 11 connects the icon representation of the previously defined components. The pulmonaryCirculation, heart and systemicCirculation components are defined as *replaceable* which allows them



Figure 7. Model of systemic arteries in OPENMODELICA.



Figure 8. Model of systemic peripherals in OPENMODELICA.



Figure 9. Model of systemic veins in OPENMODELICA.



Figure 11. Model of circulation using previously defined components in OPENMODELICA.



Figure 10. Model of systemic circulation in OPENMODELICA.

to be replaced by different implementation for future experiments and reuse as seen in the text view of the model:

```
model Hemodynamics "model of hemodynamics by Burkhoff"

  replaceable PulmonaryCirculation pulmonaryCirculation;
  PressureConst Pth(k = -533.28954966);
  FrequencyConst HeartRate(k=1.1666666666667);
  replaceable Heart heart;
  replaceable systemicCirculation systemicCirculation;
equation
...
end Hemodynamics;
```

2.2. Hemodynamics by Burkhoff et al.

The model of hemodynamics in Modelica derived from description by Burkhoff et al [4, 5] follows similar concept.

ElasticVesselWithStressedVolume component enhances ElasticVessel in the way it computes stressed volume for further feedback control in elastance.

```
model ElasticVesselWithSVandP
  extends
    Physiolibrary.Hydraulic.Components.ElasticVessel;
  VolumeOutput StressedVolume;
  PressureOutput pressure;
equation
  StressedVolume = volume - zpv;
  pressure = q_in.pressure;
end ElasticVesselWithSVandP;
```

The variable ventricular and atrial elastance are defined slightly different as in the previous model.



EtTimingBurkhoff component has input signals Tmax,TO,Tau and outputs relative elasticity combined nonlinear empirical sinus and exponential empirical equation.

```
model EtTimingBurkhoff
  TimeInput Tmax "duration of systole ";
  TimeInput T0 "elapsed time from the systole";
  TimeInput Tau "time constant of myocardial relax.";
  RealOutput Et0 "Relative elastance value"
equation
  if T0 <= 3 / 2 * Tmax then
    Et0 = 0.5 * (sin(Modelica.Constants.pi
    / Tmax * T0 - Modelica.Constants.pi / 2) + 1);
  else
    Et0 = 0.5 * exp(-(T0 - 3 * Tmax / 2) / Tau);
  end if;
end EtTimingBurkhoff;
```



HeartElastanceBurkhoff component has input signals relative elasticity Et0 generated by previous component,Vs,Alpha,Beta,Ees, outputs elasticity by empirical equation:

$$Et = (1 - Et0) * Beta * \frac{e^{Alpha*Vs} - 1}{Vs} + Et0 * Ees$$

```
model HeartElastanceBurkhoff
  RealInput Et0 "Relative elastance value ";
  VolumeInput Vs "Stressed volume";
  RealInput Alpha "coef. of end diastolic curve"
  RealInput Beta "coef. of end diastolic curve";
  HydraulicElastanceInput Ees "c.end-systolic el.";
  HydraulicElastanceOutput Et "elasticity";
equation
    Et = (1 - Et0) * Beta
         * (exp(Alpha * Vs * 1e+6) - 1) / Vs
         * 133.322368 + Et0 * Ees; //1e+6 and 133.32 is
end HeartElastanceBurkhoff;
```



HeartIntervals component computes times of atrial ventricular systole and is similar as in the previous model.

```
model HeartIntervals
  discrete Physiolibrary.Types.Time HP(start = 0)
    "heart period - duration of cardiac cycle";
  Boolean b(start = false);
  FrequencyInput HR;
  TimeOutput Tasyst, Tvsyst,T0,EDTV;
  TimeInput AVDelay "AV interval";
equation
  b = time - pre(T0) >= pre(HP);
  when {initial(),b} then
    T0 = time;
    HP = 1 / HR;
    EDTV = T0 + AVDelay;
  end when;
  Tasyst = time - T0;
  if time - T0 < AVDelay then
    Tvsyst = 0;
  else
    Tvsyst = time - (T0 + AVDelay);
  end if;
end HeartIntervals;
```

Model of side of the heart on figure 12 slightly differs from the previous model with a feedback control of stressed volume from elastic compartments to its elasticity.

Left and right heart inherits the behavior from the side of the heart and defines it's own constants.

```
model LeftHeart
  extends Physiolibrary.Icons.LeftHeart;
  extends SideOfHeart(
    xAEes(k=33330596.85375),
    xAAlpha(k=0.04),
    xABeta(k=0.3),
```



Figure 12. Abstract model of side of the heart component in OPENMODELICA.

```
    xVEes(k=205316476.6191),
    xVAlpha(k=0.024),
    xVBeta(k=0.34),
    VxAU(k=5e-06),
    CxABackflow(k=0),
    xtv(k=333305.9685375),
    VxVU(k=5e-06),
    CxVBackflow(k=0),
    xR(k=266644.77483),
    Atrium(volume_start=4e-05),
    Ventricle(volume_start=6e-05));
end LeftHeart;
```

```
model RightHeart
  extends Physiolibrary.Icons.RightHeart;
  extends SideOfHeart(
    xAAlpha(k=0.04),
    xABeta(k=0.3),
    xAEes(k=26664477.483),
    xVEes(k=50662507.2177),
    xVAlpha(k=0.024),
    xVBeta(k=0.34),
    VxAU(k=5e-06),
    CxABackflow(k=0),
    xtv(k=333305.9685375),
    VxVU(k=5e-06),
    CxVBackflow(k=0),
    xR(k=266644.77483),
    Atrium(volume_start=4e-05),
    Ventricle(volume_start=6e-05));
end RightHeart;
```

Encapsulating the left heart, right heart together with the control mechanism influencing the variable elastances of both atria and ventricles and with pressure flow measurement is in figures 13 and 14.

Pulmonary and systemic circulation are modeled similar way as in previous model. In contrast to original work of Burkhoff et al. we added an influence of intrathoracic pressure, which is present in previous model on figures 15 and 16.

We can connect the existing parts and make similar model as previous. However as we would like to utilize the features of object-oriented approach we reuse the existing Hemodynamics model and *redeclare* systemicCirculation,heart and pulmonaryCirculation with new instances



Figure 13. Model of HeartElasticities component in OPENMODELICA.



Figure 14. Model of Heart connecting left and right heart with control mechanism in OPENMODELICA.



Figure 15. Model of pulmonary circulation component in OPENMODELICA.



Figure 16. Model of systemic circulation component in OPENMODELICA.

as seen on the textual view (or generated diagram view which looks same as previous model, figure 17):

```
model HemodynamicsBurkhoff
  extends MeursModel.Models.Hemodynamics(
    redeclare BurkhoffModel.Parts.Heart heart,
    redeclare BurkhoffModel.Parts.PulmonaryCirculation
                                  pulmonaryCirculation,
    redeclare BurkhoffModel.Parts.SystemicCirculation
                                  systemicCirculation);
end HemodynamicsBurkhoff;
```



Figure 17. Model diagram of circulation in OPENMODELICA. Component instances used from definition of Burkhoff model.

3. Results

The models was tested in normal physiological and selected patophysiological condition. See the tables 1 and 2 for values of parameters of both models.

The pressure dynamics is shown on figure 18 and 19 during one cardiac cycle.

To simulate changing blood volume, we may extend current model by blood volume measurement and catheter, which may pump blood into or out from the circulation. The enhanced model is on figure 20 and defines additional equation to compute volume:

```
model Hemodynamics_with_catheter
  extends Models.Hemodynamics;
  VolumeOutput volume;
  HydraulicPort_a catheter;
equation
   volume =
        pulmonaryCirculation.PulmonaryArteries.volume+
        pulmonaryCirculation.PulmonaryVeins.volume+
        ...
end Hemodynamics_with_catheter;
```



Blood intake or outtake component will generate the flow based on the desired flow value from it's input signal.



Figure 18. Pressure in left atrium, ventricle and aorta and volume of ventricle during one cardiac cycle. Simulation of Meurs model in normal condition.



Figure 19. Pressure in left atrium, ventricle and aorta and volume of ventricle during one cardiac cycle. Simulation of Burkhoff model in normal condition.



Figure 20. Circulation model diagram enhanced by catheter and volume output.



Figure 21. Circulation model connected with volume control mechanism via catheter, which adds or removes blood from circulation based on the desired volume.

```
model BloodTransfusionHemorrhage
  HydraulicPort_b port_b;
  VolumeFlowRateInput volumeflowrate;
equation
    port_b.q = - volumeflowrate;
end BloodTransfusionHemorrhage;
```

Volume control computes desired flow from current blood volume and desired volume which may come from user input.

```
model VolumeControl "control total blood volume"
  parameter Time flowtime=0.1;
  VolumeInput volume;
  VolumeInput desiredVolume;
  VolumeFlowRateOutput volumeflowrate;
equation
  volumeflowrate = (desiredVolume-volume)/flowtime;
end VolumeControl;
```

External blood volume control mechanism added into circulation (see model in figure 21) simulate hemodynamics with different blood volumes. The same extension is done for model by Meurs as well as for model by Burkhoff. Ventrical pressure - volumetric diagram in normal condition (5.15 l) and when they have low blood volume (4.15 l) or high blood volume (6.15 l) is shown in figure 22 and 23.

4. Discussion

The implementation presented within this paper shows a way how to approach the modeling process from basic components to complex one. It is not dogmatic best practices, but in summary we use on detailed level a textual form with set of equations written in Modelica syntax e.g. models of AtrialElastance or HeartIntervals.

For higher level components we use rather the Modelica diagram - connection gives trivial equality between lower level properties of components - and rather structure is presented. Where combination of both is used it's recommended to add equation description into the appropriate icon in diagram view to keep the information, that some equations are defined behind the graphical diagram (see the model of enhanced circulation, volume is defined as



Figure 22. Ventricular pressure-volume curve for normal condition and low blood and high blood volume for model by Meurs.



Figure 23. Ventricular pressure-volume curve for normal condition and low blood and high blood volume for model by Burkhoff.

one equation). We believe that this approach keeps understandability on higher level (the model diagram fig.11 is according to the top-level structure on fig.3). In the same time the preciseness in used mathematic equations is kept on detailed level.

For identical components we define an abstract model. Concrete instances inherites the behavior and redefines the parameter values. To redefine not only values of parameters, but implementation of structures we us "replaceable" items, which are "redeclared", as shown in case of model of Burkhoff reusing the circulation structure and redeclaring instances of heart, pulmonary and systemic circulation with own implementation. This type of inheritation and polymorphism might not be adequate for all cases and where it brings more troubles on understandability the models are rather modeled using PHYSILOBRARY structures rather than to define abstract one via inheritance redefine concrete instances (see implementation of pulmonary circulation, systemic circulation).

The technique of replacing compatible structure with new implementation might be used to test new particular sub models verifying specific phenomena, but reusing the existing parts of the bigger model. E.g. Pironet et al. published an alternative to time-varying elastance theory derived from sarcomere behavior for left atrium and left ventricle, they tested this new theory with the rest of the validated models[22] and such model in Modelica might be introduced in a future work without the need to reimplement the known parts.

In most cases we define the model parameters (Const in diagram) as close as possible to the target component. However, it is possible to define parameters of models in a special model on higher level and propagate the values through several levels to the target component. The so-called "bus connector" is recommended to use to prevent mash of connection lines in model diagram.

5. Summary

This paper introduces object oriented and acausal modeling methods on exemplar implementation of current model of cardiovascular hemodynamic. Two exemplar implementation of hemodynamics of cardiovascular system was shown and an enhancement was made to achieve virtual experiments of changing total blood volume and see effects on non regulated hemodynamics of cardiovascular system. Model hierarchy keeps the exactness on detailed level together with high level overview which is key to understand the principle on each level. We believe that the described methods will lead to higher understanding and reuse in scientific and educational community and will lead to introduce a repository of models in MODELICA language to contribute to current physiome projects.

References

[1] A. C. Guyton, T. G. Coleman, H. J. Granger, Circulation: overall regulation., Annual review of physiology 34 (1972) 13–46. doi:10.1146/annurev.ph.34.030172.000305.
URL http://www.ncbi.nlm.nih.gov/pubmed/4334846

[2] R. L. Hester, A. J. Brown, L. Husband, R. Iliescu, D. Pruett, R. Summers, T. G. Coleman, Hummod: a modeling environment for the simulation of integrative human physiology, Frontiers in physiology 2.

[3] R. L. Hester, R. Iliescu, R. Summers, T. G. Coleman, Systems biology and integrative physiological modelling, The Journal of physiology 589 (5) (2011) 1053–1060.

[4] W. P. Santamore, D. Burkhoff, Hemodynamic consequences of ventricular interaction as assessed by model analysis., The American journal of physiology 260 (1 Pt 2) (1991) H146–H157.

[5] D. Burkhoff, J. V. Tyberg, Why does pulmonary venous pressure rise after onset of LV dysfunction: a theoretical analysis., The American journal of physiology 265 (5 Pt 2) (1993) H1819–H1828.

[6] Harvi - interactive simulation-based digital textbook of cardiovascular physiology and hemodynamics, http://www.pvloops.com/, web accessed: April 2014.

[7] J. A. Goodwin, W. L. van Meurs, C. D. Sá Couto, J. E. W. Beneken, S. A. Graves, A model for educational simulation of infant cardiovascular physiology., Anesthesia and analgesia 99 (6) (2004) 1655–1664, table of contents. doi:10.1213/01.ANE.0000134797.52793.AF.

[8] C. D. Sá Couto, W. L. van Meurs, J. A. Goodwin, P. Andriessen, A Model for Educational Simulation of Neonatal Cardiovascular Pathophysiology, Simulation in Healthcare 1 (Inaugural) (2006) 4–12.

[9] W. van Meurs, Modeling and Simulation in Biomedical Engineering: Applications in Cardiorespiratory Physiology, McGraw-Hill Professional, 2011.

[10] Jsim: Java-based simulation platform for data analysis, http://www.physiome.org/jsim, web: 24.2.2014.

[11] CellML: Cell Modeling Markup Language, http://www.cellml.org, web: 24.2.2014.

[12] J. F. de Canete, P. del Saz-Orozco, D. Moreno-Boza, E. Duran-Venegas, Object-oriented modeling and simulation of the closed loop cardiovascular system by using SIMSCAPE., Computers in biology and medicine 43 (4) (2013) 323–33. doi:10.1016/j.compbiomed.2013.01.007.

[13] J. Fernandez de Canete, J. Luque, J. Barbancho, V. Munoz, Modelling of long-term and short-term mechanisms of arterial pressure control in the cardiovascular system: An object-oriented approach., Computers in biology and medicine 47 (2014) 104–12. doi:10.1016/j.compbiomed.2014.01.006.

[14] J. Kofránek, J. Rusz, Restoration of Guyton's diagram for regulation of the circulation as a basis for quantitative physiological model development., Physiological research / Academia Scientiarum Bohemoslovaca 59 (6) (2010) 897–908.
URL http://www.ncbi.nlm.nih.gov/pubmed/20533860

[15] J. Kofranek, M. Matejak, P. Privitzer, Hummod-large scale physiological models in modelica, in: Proceedings 8th Modelica Conference, Dresden, Germany, 2011, pp. 713–724.

[16] J. Kofránek, M. Mateják, P. Privitzer, M. Tribula, T. Kulhánek, J. Šilar, R. Pecinovský, Hummod-golem edition: large scale model of integrative physiology for virtual patient simulators, in: Proceedings of World Congress in Computer Science 2013 (WORLDCOMP'13), International Conference on Modeling, Simulation and Visualisation Methods (MSV'13), 2013, pp. 182–188.

[17] Modelica and the Modelica Association, https://modelica.org, accessed: 2013-12-01.

[18] P. Fritzson, P. Bunus, Modelica-a general object-oriented language for continuous and discrete-event system modeling and simulation, in: Simulation Symposium, 2002. Proceedings. 35th Annual, IEEE, 2002, pp. 365–380.

[19] M. M. Tiller, Patterns and Anti-Patterns in Modelica, Proceedings Modelica 2008 Conference (2008) 647–655.

[20] P. Fritzson, P. Aronsson, A. Pop, H. Lundvall, K. Nystrom, L. Saldamli, D. Broman, A. Sandholm, Openmodelica-a free open-source environment for system modeling, simulation, and teaching, in: Computer Aided Control System Design, 2006 IEEE International Conference on Control Applications, 2006 IEEE International Symposium on Intelligent Control, 2006 IEEE, IEEE, 2006, pp. 1588–1595.

[21] M. Mateják, T. Kulhánek, J. Šilar, P. Privitzer, F. Ježek, J. Kofránek, Physiolibrary -modelica library for physiology, in: Proceedings 10th Modelica Conference, March 10-12, 2014, Lund, Sweden, 2014, pp. 499–505.
URL http://dx.doi.org/10.3384/ecp14096499

[22] A. Pironet, P. C. Dauby, S. Paeme, S. Kosta, J. G. Chase, T. Desaive, Simulation of left atrial function using a multi-scale model of the cardiovascular system., PloS one 8 (6) (2013) e65146. doi:10.1371/journal.pone.0065146.
URL http://dx.plos.org/10.1371/journal.pone.0065146

parameter	description	value unit	value in SI unit SI unit	ref.
MeursModel.Models.Hemodynamics				
Pth	intrathoracic pressure	-4 mmHg	-5.33×10^{2} Pa	[7]
HeartRate	normal heart rate	72 beats per min	1.20 Hz	[7]
Pext	extrathoracic pressure	0 mmHg	0 Pa	*
MeursModel.Parts.LeftHeart				
VxAU	unstressed volume of left atrium	30 ml	3×10^{-5} m^3	[7]
VxVU	unstressed volume of left ventricle	60 ml	6×10^{-5} m^3	
RxAOutflow	resistance of atrioventricular valve	0.003 mmHg.s/ml	4.00×10^{5} Pa.s/m^3	
RxVOutflow	resistance of ventriculoartery valve	0.008 mmHg.s/ml	1.07×10^{6} Pa.s/m^3	
ventricularElastance.EMIN	diastolic elastance of ventricle	0.09 mmHg/ml	1.20×10^{7} Pa/m^3	
ventricularElastance.EMAX	systolic elastance of ventricle	4 mmHg/ml	5.33×10^{8} Pa/m^3	
atrialElastance.EMIN	diastolic elastance of atrium	0.12 mmHg/ml	1.60×10^{7} Pa/m^3	
atrialElastance.EMAX	systolic elastance of atrium	0.28 mmHg/ml	3.73×10^{7} Pa/m^3	
atrium.volume_start	initial volume of atrium	40 ml	4×10^{-5} m^3	
ventricle.volume_start	initial volume of left ventricle	130 ml	1.30×10^{-4} m^3	
MeursModel.Parts.RightHeart				
VxAU	unstressed volume of right atrium	30 ml	3×10^{-5} m^3	[7]
VxVU	unstressed volume of right ventricle	40 ml	4×10^{-5} m^3	
RxAOutflow	resistance of atrioventricular valve	0.003 mmHg.s/ml	4.00×10^{5} Pa.s/m^3	
RxVOutflow	resistance of ventriculoartery valve	0.003 mmHg.s/ml	4.00×10^{5} Pa.s/m^3	
ventricularElastance.EMIN	diastolic elastance of ventricle	0.057 mmHg/ml	7.60×10^{6} Pa/m^3	
ventricularElastance.EMAX	systolic elastance of ventricle	0.49 mmHg/ml	6.53×10^{7} Pa/m^3	
atrialElastance.EMIN	diastolic elastance of atrium	0.05 mmHg/ml	6.67×10^{6} Pa/m^3	
atrialElastance.EMAX	systolic elastance of atrium	0.15 mmHg/ml	2.00×10^{7} Pa/m^3	
atrium.volume_start	initial volume of right atrium	40 ml	4×10^{-5} m^3	
ventricle.volume_start	initial volume of right ventricle	130 ml	1.30×10^{-4} m^3	
MeursModel.Parts.PulmonaryCirculation				
RPP	resistance of pulmonary arteries	0.11 mmHg.s/ml	1.47×10^{7} Pa.s/m^3	
VPAU	unstressed volume of pulmonary arteries	50 ml	5×10^{-5} m^3	
EPA	elastance of pulmonary arteries	0.233 mmHg/ml	3.11×10^{7} Pa/m^3	
VPVU	unstressed volume of pulmonary veins	350 ml	3.50×10^{-4} m^3	
EPV	elastance of pulmonary veins	0.0455 mmHg/ml	6.07×10^{6} Pa/m^3	
RLAIN	resistance of pulmonary veins	0.003 mmHg.s/ml	4.00×10^{5} Pa.s/m^3	
PulmonaryArteries.volume_start	initial volume of pulmonary arteries	100 ml	1×10^{-4} m^3	
PulmonaryVeins.volume_start	initial volume of pulmonary veins	600 ml	6×10^{-4} m^3	
MeursModel.Parts.SystemicPeripheralVessels				
RTA	resistance of small venules	0.2 mmHg.s/ml	2.67×10^{7} Pa.s/m^3	
TUV	unstressed volume of small venules	185 ml	1.85×10^{-4} m^3	
EET	elastance of small venules	0.262 mmHg/ml	3.49×10^{7} Pa/m^3	
RSP	peripheral resistance	0.8 mmHg.s/ml	1.07×10^{8} Pa.s/m^3	
MeursModel.Parts.SystemicArteries				
VETHAU	unstressed volume of extrathoracic arteries	370 ml	3.70×10^{-4} m^3	
EETHA	elastance of extrathoracic arteries	0.556 mmHg/ml	7.41×10^{7} Pa/m^3	
RETHA	resistance of extrathoracic arteries	0.06 mmHg.s/ml	8.00×10^{6} Pa.s/m^3	
VITHAU	unstressed volume of intrathoracic arteries	140 ml	1.40×10^{-4} m^3	
EITHA	elastance of intrathoracic arteries	1.43 mmHg/ml	1.91×10^{8} Pa/m^3	
MeursModel.Parts.SystemicVeins				
RRAIN	resistance of central veins	0.003 mmHg.s/ml	4.00×10^{5} Pas.s/m^3	
VITHVU	unstressed volume of intrathoracic veins	1190 ml	1.19×10^{-3} m^3	
EITHV	elastance of intrathoracic veins	0.0182 mmHg/ml	2.43×10^{6} Pa/m^3	
RETHV	resistance of extrathoracic veins	0.09 mmHg.s/ml	1.20×10^{7} Pas.s/m^3	
VETHVU	unstressed volume of extrathoracic veins	1000 ml	1×10^{-3} m^3	
EETHV	elastance of intrathoracic veins	0.0169 mmHg/ml	2.25×10^{6} Pa/m^3	

Table 1. Initial values of state variables and parameters of the cardiovascular models by Meurs in Modelica per components. Parameters refered by * were added for further simulation scenarios.

parameter	description	value unit	value in SI unit	SI unit	ref.
BurkhoffModel.Models.Hemodynamics					
Pth	intrathoracic pressure	-4 mmHg	-5.33×10^{2}	Pa	[7]
HeartRate	normal heart rate	72 beats per min	1.20	Hz	[7]
Pext	extrathoracic pressure	0 mmHg	0	Pa	*
BurkhoffModel.Parts.LeftHeart					
VxAU	unstressed volume of left atrium	30 ml	3×10^{-5}	m^3	[7]
VxVU	unstressed volume of left ventricle	60 ml	6×10^{-5}	m^3	
RxAOutflow	resistance of atrioventricular valve	0.003 mmHg.s/ml	4.00×10^{5}	Pa.s/m^3	
RxVOutflow	resistance of ventriculoartery valve	0.008 mmHg.s/ml	1.07×10^{6}	Pa.s/m^3	
ventricularElastance.EMIN	diastolic elastance of ventricle	0.09 mmHg/ml	1.20×10^{7}	Pa/m^3	
ventricularElastance.EMAX	systolic elastance of ventricle	4 mmHg/ml	5.33×10^{8}	Pa/m^3	
atrialElastance.EMIN	diastolic elastance of atrium	0.12 mmHg/ml	1.60×10^{7}	Pa/m^3	
atrialElastance.EMAX	systolic elastance of atrium	0.28 mmHg/ml	3.73×10^{7}	Pa/m^3	
atrium.volume_start	initial volume of atrium	40 ml	4×10^{-5}	m^3	
ventricle.volume_start	initial volume of left ventricle	130 ml	1.30×10^{-4}	m^3	
BurkhoffModel.Parts.RightHeart					
VxAU	unstressed volume of right atrium	30 ml	3×10^{-5}	m^3	[7]
VxVU	unstressed volume of right ventricle	40 ml	4×10^{-5}	m^3	
RxAOutflow	resistance of atrioventricular valve	0.003 mmHg.s/ml	4.00×10^{5}	Pa.s/m^3	
RxVOutflow	resistance of ventriculoartery valve	0.003 mmHg.s/ml	4.00×10^{5}	Pa.s/m^3	
ventricularElastance.EMIN	diastolic elastance of ventricle	0.057 mmHg/ml	7.60×10^{6}	Pa/m^3	
ventricularElastance.EMAX	systolic elastance of ventricle	0.49 mmHg/ml	6.53×10^{7}	Pa/m^3	
atrialElastance.EMIN	diastolic elastance of atrium	0.05 mmHg/ml	6.67×10^{6}	Pa/m^3	
atrialElastance.EMAX	systolic elastance of atrium	0.15 mmHg/ml	2.00×10^{7}	Pa/m^3	
atrium.volume_start	initial volume of right atrium	40 ml	4×10^{-5}	m^3	
ventricle.volume_start	initial volume of right ventricle	130 ml	1.30×10^{-4}	m^3	
BurkhoffModel.Parts.PulmonaryCirculation					
RPP	resistance of pulmonary arteries	0.11 mmHg.s/ml	1.47×10^{7}	Pa.s/m^3	
VPAU	unstressed volume of pulmonary arteries	50 ml	5×10^{-5}	m^3	
EPA	elastance of pulmonary arteries	0.233 mmHg/ml	3.11×10^{7}	Pa/m^3	
VPVU	unstressed volume of pulmonary veins	350 ml	3.50×10^{-4}	m^3	
EPV	elastance of pulmonary veins	0.0455 mmHg/ml	6.07×10^{6}	Pa/m^3	
RLAIN	resistance of pulmonary veins	0.003 mmHg.s/ml	4.00×10^{5}	Pa.s/m^3	
PulmonaryArteries.volume_start	initial volume of pulmonary arteries	100 ml	1×10^{-4}	m^3	
PulmonaryVeins.volume_start	initial volume of pulmonary veins	600 ml	6×10^{-4}	m^3	
BurkhoffModel.Parts.SystemicPeripheralVessels					
RTA	resistance of small venules	0.2 mmHg.s/ml	2.67×10^{7}	Pa.s/m^3	
TUV	unstressed volume of small venules	185 ml	1.85×10^{-4}	m^3	
EET	elastance of small venules	0.262 mmHg/ml	3.49×10^{7}	Pa/m^3	
RSP	peripheral resistance	0.8 mmHg.s/ml	1.07×10^{8}	Pa.s/m^3	
RRAIN	resistance of central veins	0.003 mmHg.s/ml	4.00×10^{5}	Pas.s/m^3	
VITHVU	unstressed volume of intrathoracic veins	1190 ml	1.19×10^{-3}	m^3	
EITHV	elastance of intrathoracic veins	0.0182 mmHg/ml	2.43×10^{6}	Pa/m^3	
RETHV	resistance of extrathoracic veins	0.09 mmHg.s/ml	1.20×10^{7}	Pas.s/m^3	
VETHVU	unstressed volume of extrathoracic veins	1000 ml	1×10^{-3}	m^3	
EETHV	elastance of intrathoracic veins	0.0169 mmHg/ml	2.25×10^{6}	Pa/m^3	

Table 2. Initial values of state variables and parameters of the cardiovascular models by Burkhoff in Modelica per components. Parameters refered by * were added for further simulation scenarios.